
Computer Systems 
 

Understanding Stack Usage for Function Calls in X86 
 
Essential Concepts – Review: 
 
1. The stack is a region of your executable program used to facilitate function calls. It stores 

function arguments, local variables and return addresses, among other things. 
2. When an executable program is invoked, the operating system loads it from the disk into 

memory and then begins execution. So when we talk about the stack, we talk about memory.  
3. In Intel X86 assembly, square brackets indicate memory operands: to access data in memory, 

the memory address of that data is enclosed within square brackets. For example, the X86 
instruction  

 
               mov EAX, [EBX]  
 

has the effect of moving into register EAX four memory bytes from the memory address 
stored in register EBX. Compare this with the X86 instruction 

 
               mov EAX, EBX  
 

which has the effect of copying the value of the register EBX into register EAX. 
 
What to do: 
 
In this lab we will use the debugger to examine the dynamics of the stack (allocation and de-
allocation of function call frames on the stack) during program execution.  
 
1. Copy stackdemo.c from /mnt/a/mdamian/x86 into your systems/x86 directory.  

 

 
 
2. Compile with  

 gcc -O stackdemo.c –o xstackdemo 
The first option –O (capital O) tells the compiler to generate optimized code.  
 

#include <stdio.h> 
 
int sum(int x, int y) 
{ 
  return x+y; 
} 
 
int main() 
{ 
  int result; 
  result = sum(0x11, 0x22); 
  printf("%d\n", result); 
  return 0; 
} 



3. Load xstackdemo into the debugger (gdb) and disassemble the code for main: 
 

gdb xstackdemo 
set disassembly-flavor intel 
disas main 

 
Disassembled code for main: 
 
0x080483d9 <main+10>: push   ebp 
0x080483da <main+11>: mov    ebp,esp 
0x080483dc <main+13>: push   ecx 
0x080483dd <main+14>: sub    esp,0x14 
 
0x080483e0 <main+17>: mov    DWORD PTR [esp+0x4],0x22 
0x080483e8 <main+25>: mov    DWORD PTR [esp],0x11 
0x080483ef <main+32>: call   0x80483c4 <sum> 
0x080483f4 <main+37>: mov    DWORD PTR [esp+0x4],eax 
0x080483f8 <main+41>: mov    DWORD PTR [esp],0x80484e4 
0x080483ff <main+48>: call   0x80482f4 <printf@plt> 
0x08048404 <main+53>: mov    eax,0x0 
 
0x08048409 <main+58>: add    esp,0x14 
0x0804840c <main+61>: pop    ecx 
0x0804840d <main+62>: pop    ebp 
0x0804840e <main+63>: lea    esp,[ecx-0x4] 
0x08048411 <main+66>: ret     

 
4. Next disassemble the code for sum: 
 

Disassembled code for sum: 
      
0x080483c4 <sum+0>: push   ebp 
0x080483c5 <sum+1>: mov    ebp,esp 
 
0x080483c7 <sum+3>: mov    eax,DWORD PTR [ebp+0xc] 
0x080483ca <sum+6>: add    eax,DWORD PTR [ebp+0x8] 
 
0x080483cd <sum+9>: pop    ebp 
0x080483ce <sum+10>: ret     

 
 

5. We would like to inspect the contents of the stack just before the function sum is called, so 
let us set a breakpoint at the X86 instruction calling the sum function: 
 
break *(main+32) # Set breakpoint to suspend execution 
 

6. The first thing to do before stepping through your code with the debugger is to tell the 
debugger to remind you what the next instruction to execute is: 

 
display /i $eip # What gets executed next?  



Recall that EIP is a special register in the Intel X86 architecture (called Instruction Pointer) 
that always points to (contains the address of) the next instruction to execute. 

 
7. Start executing the program:  

 
  run    
 
Execution will be suspended at the breakpoint, and the next instruction to execute (calling 
sum) will be displayed.  

 
8. Inspect the two top stack elements. What do you expect to find there?  

 
x $esp   # First argument 
x $esp+4  # Second argument 
 
Notice the use of the debugger command x (which stands for examine memory). Use this 
command whenever you wish to access memory contents (indicated by square brackets in 
the X86 instruction). In this example, we are trying to see what is on the stack, so we are 
trying to access memory contents.  
 
Recall that we are just about to call the function sum, so the arguments of sum should be at 
the top of the stack.   
 

9. Our next goal is to inspect the contents of the stack immediately following the call 
instruction, so let us place a breakpoint at the first instruction in sum and continue execution 
up to that point: 
 
break *(sum)    
cont   # continue execution up to sum 
 

10. Let us inspect again the top stack element. What do you expect to find there?  
 
x $esp   
 
Recall that the call instruction places the return address of the caller (main in this case) 
on top of the stack. What lies just below that on the stack? Let’s look: 
 
x $esp+4   
x $ebp+8   

 
Indeed, we find the arguments of sum placed on the stack just before the call instruction. 
Notice the dynamic nature of the stack – it grows and shrinks (to be noted later) during 
program execution.  
 

11. Execute the prolog of the sum function by stepping through the first two instructions: 
 
ni 
ni 
 



So what just happened? The value stored into EBP was saved on the stack, just before EBP 
was updated to point to the top of the stack. This enables the function sum to access its 
arguments relative to EBP instead of ESP (so the stack is free to grow while sum executes).  

 
x $ebp+4   # Return address for main 
x $ebp+8   # First argument of sum 
x $ebp+12   # Second argument of sum 
 
EBP will remain unchanged during the execution of sum, but ESP may change. This is why 
it is important to use EBP in accessing function arguments at this point, instead of ESP.  

 
12. Set a breakpoint at the ret instruction in sum, then continue execution up to that 

breakpoint: 
 
break *(sum+10) 
cont 

 
13. We are just about to execute the ret instruction. What do we expect to find in the register 

eax? Check it out: 
    

p $eax  # return value of sum 
 

 What do we expect to find on top of the stack? Verify your answer: 
 

x $esp 
 

Indeed, the return address of main should be at the top of the stack at this point. The ret 
instruction simply loads this address into register EIP, so that the next instruction to execute 
will be the one at that address.  
 
What’s below the return address on the stack? (Recall what was placed on the stack before 
sum was called) 
 
x $esp+4 
x $esp+8 

 
14. Execute the ret instruction, and notice the jump back to the main function: 

 
ni 
 
What’s on top of the stack now? Verify that the return address has been popped from the 
stack: 
 
x $esp 
x $esp+4 
 

15.  Complete the execution of the program: 
 

cont 


